Naval Research Laboratory Stennis Space Center, MS 39529-5004

NRL/PU/7331--98-0340

Distribution of Light Radiance Reflected from the Sea Illuminated by the Sun with Shadow: Algorithm, Program and Examples

Vladimir I. Haltrin

Naval Research Laboratory, Ocean Sciences Branch, Code 7331 Stennis Space Center, MS 39529-5004, USA Phone: 228-688-4528, fax: 228-688-5379, e-mail: <haltrin@nrlssc.navy.mil>

May 5, 1998

Approved for public release; distribution unlimited

This page intentionally left blank

CONTENTS

EXECUTIVE SUMMARY
INTRODUCTION
FORMAL METHODS7
HOMOGENEOUS SHALLOW SEA ILLUMINATED BY SUNLIGHT 7
SOLUTIONS FOR THE RADIANCE
SOLUTIONS FOR THE UPWELLING RADIANCE NEAR THE SEA SURFACE 10
HOMOGENEOUS SEA ILLUMINATED BY SUNLIGHT WITH SHADOW 10
RADIANCE ABOVE THE SEA SURFACE
APPROXIMATION FOR THE PHASE FUNCTION OF SCATTERING 13
EXAMPLES OF COMPUTATIONS
CONCLUSIONS22
ACKNOWLEDGMENTS22
REFERENCES
APPENDIX A — FORTRAN PROGRAM RADPIER 24
DISTRIBUTION LIST

This page intentionally left blank

EXECUTIVE SUMMARY

Introduction

Detailed, quantitative interpretations of images require data about the angular distribution of light reflected from a shadowed sea. This report demonstrates the solution of the two-dimensional radiative transfer problem for the sea shadowed by an opaque body such as a pier (See Figure 1). The solution to the problem requires two different theoretical approaches. The solution for the depth dependence of light radiance is obtained using the theory of Green's functions (Vladimirov, 1971). The solution to the problem in horizontal direction is obtained with the one-dimensional Fourier transform method (Morse and Feshbach, 1953).


The algorithm used in this report is based on the analytical solution to the spatially two-dimensional (Cartesian coordinates x and z) scalar radiative transfer equation in seawater. The θx axis starts on the sea surface in the center of the opaque shadow (pier) and increases towards the sun in the direction orthogonal to the axis of the pier. The θz axis increases down to the sea bottom.

Because the input parameters did not include the scattering phase function, the algorithm uses experimental results published in literature. The scattering phase function is estimated through the regression relationships which are derived by processing all fifteen experimental marine Petzold phase functions (Haltrin, 1997). This regression relationship couples the scattering phase function with the scattering coefficient and the single-scattering albedo.

The future possible development of this problem may consists of the following enhancements to the theory: (a) consideration of the reflection by the sea bottom; (b) consideration of the atmospheric scattering effects (Haltrin, 1998b, 1998c) and the elevation of the pier; (c) consideration of the sea surface roughness; (d) generalization to the non-infinite pier (a 3D-problem in space); (e) consideration of the vertical optical stratification of seawater; and (f) enhancement of the part of the theory related to the higher-order scattering effects.

Findings and Recommendations

The final equation obtained in this report expresses the upwelling sea radiance distribution as a function of the sun zenith angle, azimuth and zenith angles of viewing, the width of the pier and the inherent optical properties of the water (absorption and attenuation coefficients and the scattering phase function). Results for the actual radiance distribution of light are obtained numerically with the program RADPIER (see Appendix) using as input the *in situ* measured optical properties of seawater. The scattering phase function is estimated through the regression equations that are based on all fifteen Petzold scattering phase functions. The empirically derived scattering phase function depends on the scattering coefficient, the single-scattering albedo, and the scattering angle. The resulting radiance distributions are discussed and illustrated.



INTRODUCTION

Our goal is formulated as follows: Given a downwelling radiance distribution from the sun consistent with a long rectangular shadow, to model the water-leading radiance as a function of azimuth and zenith angles of viewing and distance in the direction normal to the long dimension of the rectangle. The derived approach should be valid for an arbitrary solar elevation.

Several serious attempts have been made to solve two- and three-dimensional radiative transfer problem (Vladimirov, 1961; Williams, 1967; Kochetkov, 1968; Kaper, 1969; Case and Hazeltine, 1970, Flateau and Stephens, 1988). Unfortunately none of these approaches are simple and convenient enough to produce practically usable equations applicable to solve our problem.

FORMAL METHODS

We start from the scalar radiative transfer theory (Chandrasekhar, 1960). Our problem is a two dimensional in space radiative transfer problem. The origin of the Cartesian coordinates is placed on the sea surface just below the center of an infinite pier. For simplicity we place the pier horizontally along the 0y axis, and the solar plane is perpendicular to the 0y axis. The 0x axis is directed horizontally to the sun and lies in the solar plane, and 0z axis is directed vertically to the bottom.

We use the Green's function method (Vladimirov, 1971) to solve the problem in a *z*-direction, and the Fourier transform method to solve an *x*-dependent part (Morse and Feshbach, 1953).

The angular distribution of radiance is split into three components: unscattered, single-scattered and multiple scattered components. Exact solutions for the first two components are obtained.

HOMOGENEOUS SHALLOW SEA ILLUMINATED BY THE SUNLIGHT

We assume that the reader is familiar with the standard terminology of hydrologic optics (see, for example, Jerlov, 1986; Shifrin 1988, Mobley, 1994).

Let us introduce the following notations: E_S is irradiance by the sun on the sea surface, $E_S^0 = E_S / \mu_S^0$ is irradiance by the sun on the surface normal to the solar rays, $\mu_S^0 = \cos Z_S \equiv \sin h_S$ is the cosine of the solar zenith angle Z_S , $h_S = 90^\circ - Z_S$ is a solar elevation angle.

The solar radiance above the sea surface is expressed by:

$$L_{\rm s}^0 = E_{\rm s}^0 \, \delta(\mu - \mu_{\rm s}^0) \, \delta(\varphi), \tag{1}$$

where $\delta(x)$ is the Dirac's delta-function (Morse and Feshbach, 1953)

$$\mu_{S} = \sqrt{1 - \sin^{2} Z_{S} / n_{w}^{2}} \tag{1a}$$

is the cosine of the light penetration angle, n_w is the seawater refraction coefficient.

If $T_S^{\downarrow} = 1 - R_F^{\downarrow}(Z_S)$ is a transmission of light by sea surface, where R_F is a Fresnel reflection coefficient of light falling from above, than the light radiance below the sea surface is:

$$L_{w}^{0}(\mu, \varphi) = E_{w}^{0} \delta(\mu - \mu_{S}) \delta(\varphi), \quad E_{w}^{0} = E_{S}^{0} T_{S}^{\downarrow}, \tag{2}$$

here E_w^0 is the solar irradiance on the surface normal to the rays just below the sea surface.

A one-dimensional scalar radiative transfer equation (Chandrasekhar, 1960) for the total light radiance can be written as:

$$\left(\mu \frac{d}{dz} + c\right) L(z, \mu, \varphi) = \frac{b}{4\pi} \int_{0}^{2\pi} d\varphi' \int_{-1}^{1} p(\cos \gamma) L(z, \mu', \varphi') d\mu', \tag{3}$$

here c = a + b is the attenuation coefficient, a is the absorption coefficient and b is the scattering coefficient of seawater, $\mu = \cos \theta$, the direction of light propagation is determined by the zenith and azimuth angles θ and φ , $p(\cos \gamma)$ is the scattering phase function, γ is the scattering angle determined from the following formula:

$$\cos \gamma = \mu \mu' + \sqrt{1 - \mu^2} \sqrt{1 - {\mu'}^2} \cos(\varphi - \varphi'). \tag{4}$$

Let us represent the total radiance of light inside the sea L as the direct light radiance L_0 , and the sum of radiances L_n that represent nth order of scattering:

$$L = \sum_{n=0}^{\infty} b^n L_n \,. \tag{5}$$

By substituting Eq. (5) into Eq. (3) we obtain the following equations for radiance components of different orders of scattering denoted by a subscript:

$$\left(\mu \frac{d}{dz} + c\right) L_0(z, \mu, \varphi) = 0, \tag{6}$$

$$\hat{D}L_n(z,\mu,\varphi) = b\hat{S}L_{n-1}(z,\mu,\varphi), \text{ or } L_n(z,\mu,\varphi) = b\hat{D}^{-1}\hat{S}L_{n-1}(z,\mu,\varphi) = b\hat{T}L_{n-1}(z,\mu,\varphi).$$
 (7)

The operators \hat{D} and \hat{S} are defined as follows:

$$\hat{D}F(z,\mu,\varphi) \equiv \left(\mu \frac{d}{dz} + c\right) F(z,\mu,\varphi),$$

$$\hat{S}F(z,\mu,\varphi) \equiv \frac{1}{4\pi} \int_{0}^{2\pi} d\varphi' \int_{-1}^{1} p(\cos\gamma) F(z,\mu',\varphi') d\mu',$$
(8)

and the following relationships are valid:

$$\hat{D}^{-1} \equiv \hat{G}, \quad \hat{T} = \hat{D}^{-1} \hat{S} \equiv \hat{G} \hat{S}. \tag{8a}$$

The operator \hat{G} in Eq. (8a) is a Green's function of Eq. (3). This operator is defined according to the following formula (Vladimirov, 1971):

$$\hat{G}F(z) = \int_{-\infty}^{+\infty} G(z - z')F(z')dz', \quad G(z) = \frac{H(z/\mu)}{|\mu|}e^{-cz/\mu}, \tag{9}$$

Here H(z) is the Heavyside's or step function $(H(z) = 1, z \ge 0; H(z) = 0, z < 0)$. (Morse and Feshbach, 1953). The total scattering operator is:

$$\hat{T}F(z,\mu,\varphi) = \frac{1}{4\pi} \int_{0}^{2\pi} d\varphi' \int_{-1}^{1} d\mu' \, p(\cos\gamma) \int_{-\infty}^{+\infty} dz' G(z-z') F(z',\mu',\varphi') \,. \tag{10}$$

Now, the total radiance is expressed as a following sum:

$$L = \sum_{n=0}^{\infty} b^n L_n, \quad L_n = b^n \hat{T}^n L_0, \quad n \ge 1, \quad L(z, \mu, \varphi) = \sum_{n=0}^{\infty} b^n \hat{T}^n L_0(z, \mu, \varphi). \tag{11}$$

Let us proceed to solve these equations.

SOLUTIONS FOR THE RADIANCE

The solution for direct (unscattered) radiance L_0 is straightforwardly obtained from the Eq. (6) with the following boundary condition on the sea surface, $L_0(0, \mu, \varphi) = L_w^0(\mu, \varphi)$:

$$L_0(z,\mu,\varphi) = E_w^0 e^{-cz/\mu_S} \delta(\mu - \mu_S) \delta(\varphi). \tag{12}$$

Equation (7) for the single-scattered radiance can be written as:

$$L_{1}(z,\mu,\varphi) = \frac{1}{4\pi} \int_{0}^{2\pi} d\varphi' \int_{-1}^{1} d\mu' \, p(\cos\gamma) \int_{-\infty}^{+\infty} dz' G(z-z') \, L_{0}(z',\mu',\varphi') \,. \tag{13}$$

By substituting Eq. (12) into Eq. (13), we have the following solution for the single-scattered radiance distribution:

$$L_{1}(z,\mu,\phi) = \frac{E_{w}^{0} p(\cos \gamma_{s})}{4\pi c |\mu|} \left[\psi_{1}(z,\mu) H(\mu) + \psi_{2}(z,\mu) H(-\mu) \right], \tag{14}$$

here

$$\psi_{1}(z,\mu) = \frac{e^{-cz/\mu_{S}} - e^{-cz/\mu}}{\frac{1}{\mu} - \frac{1}{\mu_{S}}}, \quad \psi_{1}(z,\mu_{S}) = cz e^{-cz/\mu_{S}}, \quad \psi_{1}(z,\mu)\big|_{z\to 0} = cz, \quad (15)$$

$$\psi_2(z,\mu) = \frac{e^{-cz/\mu_S}}{\frac{1}{\mu_S} + \frac{1}{|\mu|}} . \tag{16}$$

For the required precision about 15-20% it is sufficient (see Shuleykin, 1968) to calculate the next term in expansion given by Eq. (11). The explicit solution for the next term can be obtained from the following equation:

$$L_{2}(z,\mu,\varphi) = \frac{1}{4\pi} \int_{0}^{2\pi} d\varphi' \int_{-1}^{1} d\mu' \, p(\cos\gamma) \int_{-\infty}^{+\infty} dz' G(z-z') \, L_{1}(z',\mu',\varphi') \,. \tag{17}$$

with L_1 given by Eqs. (14)-(16) and G(z) given by Eq. (9). The right boundary conditions has no influence on upwelling radiance distribution, so we do not discuss them here.

SOLUTIONS FOR THE UPWELLING RADIANCE NEAR THE SEA SURFACE

Taking integrals in (17) and summing all required terms at z = 0 and $\mu < 0$, we have the following solution for the case of homogeneous illumination:

$$L(0,\mu,\varphi) = \frac{E_w^0 \,\mu_S \,b}{4\pi \,c \left(\mu_S + |\mu|\right)} \left[p(\cos\gamma_S) + \frac{b}{2\,c} \,\psi_p(\mu,\varphi) \right], \quad \mu < 0, \tag{18}$$

here

$$\cos \gamma_s = \mu \mu_s + \sqrt{1 - \mu^2} \sqrt{1 - \mu_s^2} \cos \varphi, \tag{19}$$

$$\psi_{p}(\mu, \varphi) = |\mu| \int_{0}^{1} \frac{x_{p}(\mu, \mu', \varphi)}{\mu' + |\mu|} d\mu' + \frac{\mu_{s}}{\mu_{s} + |\mu|} \int_{-1}^{0} x_{p}(\mu, \mu', \varphi) d\mu',$$
 (20)

$$x_p(\mu, \mu', \varphi) = \frac{1}{2\pi} \int_0^{2\pi} p(\cos \gamma) p(\cos \gamma'_S) d\varphi', \qquad (21)$$

$$\cos \gamma_S' = \mu' \mu_S + \sqrt{1 - {\mu'}^2} \sqrt{1 - {\mu_S}^2} \cos \varphi'. \tag{22}$$

HOMOGENEOUS SEA ILLUMINATED BY SUNLIGHT WITH SHADOW

Let us formulate a two-dimensional problem that takes into account an unhomogeneous over horizontal axis 0x illumination. The two-dimensional radiative transfer equation for the total angular radiance distribution is:

$$\left(\mu \frac{\partial}{\partial z} + \sqrt{1 - \mu^2} \cos \varphi \frac{\partial}{\partial x} + c\right) L(x, z, \mu, \varphi) = \frac{b}{4\pi} \int_{0}^{2\pi} d\varphi' \int_{-1}^{1} p(\cos \gamma) L(x, z, \mu', \varphi') d\mu', \quad (23)$$

Let us define a one-dimensional Fourier transform by the following equations:

$$f_k = \int_{-\infty}^{+\infty} f(x)e^{-ikx}dx$$
, $f(x) = \frac{1}{2\pi} \int_{-\infty}^{+\infty} f_k e^{ikx}dk$. (24)

After taking a Fourier transform Eq. (23) transfers to the one-dimensional equation for the Fourier amplitude $L_k(z, \mu, \varphi)$:

$$\left(\mu \frac{d}{dz} + \dot{c}\right) L_k(z, \mu, \varphi) = \frac{b}{4\pi} \int_0^{2\pi} d\varphi' \int_{-1}^1 p(\cos\gamma) L_k(z, \mu', \varphi') d\mu', \tag{25}$$

here

$$\dot{c} = c(1 + ik\tau), \quad \tau = \sqrt{1 - \mu^2} \cos \varphi / c. \tag{26}$$

Equation (25) may be formally obtained from Eq.(3) by replacing the real extinction coefficient c by the complex value \dot{c} given by Eq. (26). It means that the solution of Eq. (24) can be obtained from the solution (18) by replacing all x-dependent values by their Fourier transforms and all instances of the extinction coefficient c by the \dot{c} . Taking this into account, we have the following solution to Eq. (25):

$$L_k(0,\mu,\varphi) = \frac{E_{wk} \mu_S b}{4\pi \dot{c} \left(\mu_S + |\mu|\right)} \left[p(\cos \gamma_S) + \frac{b}{2\dot{c}} \psi_p(\mu,\varphi) \right]. \tag{27}$$

Now we only need to calculate a Fourier transform E_{wk} of the x-dependent radiance distribution

$$E_{wk} = \int_{-\infty}^{+\infty} E_w(x)e^{-ikx}dx.$$
 (28)

In our case the solar illumination incorporates a shadowing at $-w \le x < w$, where w is a half width of a shadow. The angular-space distribution of the light just under the sea surface is:

$$L_w^0(\mu, \varphi) = E_w(x)\delta(\mu - \mu_S)\delta(\varphi), \tag{29}$$

$$E_{w}(x) = E_{w}^{0} \left[1 - H(w - |x|) \right] \equiv E_{w}^{0} \left\{ 1 - 0.5 \left[sign(w - x) + sign(w + x) \right] \right\}, \tag{30}$$

here sign(x) = |x|/x, i.e., sign(x) = 1, x > 0, sign(x) = -1, x < 0.

The Fourier transform of the surface illumination is:

$$E_{wk} = E_w^0 \left[2\pi \, \delta(k) - \frac{2}{k} \sin(wk) \right]. \tag{31}$$

The *x*-dependent radiance distribution just below the sea surface is:

$$L(x,0,\mu,\varphi) = \frac{\mu_{S} \omega_{0} E_{w}^{0}}{4\pi \left(\mu_{S} + |\mu|\right)} \int_{-\infty}^{+\infty} \frac{\delta(k) - \sin(wk)/(k\pi)}{(1 + ik\tau)} \left[p(\cos\gamma_{S}) + \frac{\omega_{0}}{2(1 + ik\tau)} \psi_{p}(\mu,\varphi) \right] e^{ikx} dk . (32)$$

By taking appropriate integrals we have the following solution for the upwelling radiance distribution just under the sea surface:

$$L(x,0,\mu,\varphi) = \frac{\omega_0 \,\mu_S \,E_w^0}{4\pi \left(\mu_S + |\mu|\right)} \left\{ \left[1 - F_1(w,x,\tau)\right] \,p(\cos\gamma_S) + \frac{\omega_0}{2} \left[1 - F_2(w,x,\tau)\right] \psi_p(\mu,\varphi) \right\}, (33)$$

$$F_{1}(w,x,\tau) = \frac{1}{2} \left[\operatorname{sign}(w+x) \left(1 - e^{-\left| \frac{w+x}{\tau} \right|} \right) + \operatorname{sign}(w-x) \left(1 - e^{-\left| \frac{w-x}{\tau} \right|} \right) \right]$$

$$+ \frac{1}{2} \operatorname{sign} \tau \left(e^{-\left| \frac{w-x}{\tau} \right|} - e^{-\left| \frac{w+x}{\tau} \right|} \right), \qquad \tau = \frac{\sqrt{1 - \mu^{2} \cos \varphi}}{c} ,$$

$$(34)$$

$$F_{2}(w,x,\tau) = F_{1}(w,x,\tau) + \frac{1}{2\tau} \left[|w - x|e^{-\left|\frac{w-x}{\tau}\right|} - |w + x|e^{-\left|\frac{w+x}{\tau}\right|} \right] - \frac{1}{2|\tau|} \left[(w-x)e^{-\left|\frac{w-x}{\tau}\right|} + (w+x)e^{-\left|\frac{w+x}{\tau}\right|} \right],$$
(35)

Far from the shadow in each direction (at |w-x|, $|w+x| >> |\tau|$) $F_1 \to 0$ and $F_2 \to 0$ and Eq. (33) coincides with the equation for upwelling radiance (18) for homogeneous illumination.

Equations (33)-(40) include functions ψ_p and x_p , given by Eqs. (20)-(21), that involve integrations over angular variables μ and φ . In order to simplify these expressions let us substitute phase functions inside integrals in Eqs (20)-(21) by their transport equivalents:

$$p(\cos \gamma) \rightarrow 2B + 2(1 - 2B)\delta(\mu - \mu')\delta(\varphi - \varphi'),$$
 (36)

$$p(\cos \gamma_S') \to 2B + 2(1 - 2B)\delta(\mu' - \mu_S)\delta(\varphi' - \varphi_S), \quad \varphi_S = 0, \tag{37}$$

$$B = 0.5 \int_{-1}^{0} p(\mu) d\mu. \tag{38}$$

In this case after appropriate integrations we have:

$$x_p(\mu, \mu', \varphi) = 2B \,\overline{p}(\mu, \mu') + 2(1 - 2B) \,p(\cos \gamma_s),$$
 (39)

$$\overline{p}(\mu, \mu') = \frac{1}{2\pi} \int_{0}^{2\pi} p(\cos \gamma) d\varphi', \tag{40}$$

$$\psi_{p}(\mu, \varphi) = 4B(1 - 2B)|\mu|\log\frac{1 + |\mu|}{|\mu|} + \frac{4[B(1 - B)\mu_{S} + (1 - 2B)^{2}|\mu|]}{\mu_{S} + |\mu|}.$$
 (41)

RADIANCE ABOVE THE SEA SURFACE

In order to calculate radiance distribution above the sea surface as a function of viewing angles, we have to do the following: a) we should take into account the transmission by sea-air interface from below by multiplying result by the transmission coefficient:

$$T_{\rm s}^{\uparrow}(\mu) = 1 - R_{\rm F}^{\uparrow}(\mu); \tag{42}$$

b) we should express cosine $|\mu|$ through the zenith viewing angle θ :

$$|\mu| = \sqrt{1 - \sin^2 \theta / n_w^2}; \tag{43}$$

It means that we should make the following two substitutions in Eq. (33):

$$E_w^0 \to E_S^0 T_S^{\downarrow}(Z_S) T_S^{\uparrow}(\mu) \equiv E_S \frac{\left[1 - R_F^{\downarrow}(Z_S)\right] \left[1 - R_F^{\uparrow}(\theta)\right]}{\cos Z_S}, \tag{44}$$

and

$$\mu \to \sqrt{1 - \sin^2 \theta / n_w^2} \,; \tag{45}$$

here R_F^{\downarrow} and R_F^{\uparrow} are Fresnel reflection coefficients of sea surface for the light falling from above and below:

$$R_F^{\downarrow}(\mu) = \frac{1}{2} \left[\left(\frac{\mu - n_w \mu_w}{\mu + n_w \mu_w} \right)^2 + \left(\frac{n_w \mu - \mu_w}{n_w \mu + \mu_w} \right)^2 \right], \quad \mu_w = \frac{\sqrt{n_w^2 - 1 + \mu^2}}{n_w}, \tag{46}$$

$$R_F^{\uparrow} = \frac{1}{2} \left[\left(\frac{n_w \cos \theta - \mu_a}{n_w \cos \theta + \mu_a} \right)^2 + \left(\frac{\cos \theta - n_w \mu_a}{\cos \theta + n_w \mu_a} \right)^2 \right], \quad \mu_a = \sqrt{n_w^2 \sin^2 \theta - 1}. \tag{47}$$

APPROXIMATION FOR THE PHASE FUNCTION OF SCATTERING

The backscattering coefficient of light in seawater is connected with the average cosine of light distribution by the following equations (Haltrin, 1985, 1998a; Haltrin and Kattawar, 1993):

$$b_B = \frac{ax}{1-x}, \quad x = \frac{(1-\overline{\mu}^2)^2}{1+4\overline{\mu}^2-\overline{\mu}^4},$$
 (48)

This regression is derived by the author from the experimental results by Timofeyeva (1971):

$$\overline{\mu} = y(2.6178398 + y(-4.6024180 + y(9.0040600 + y(-14.59994 + y(14.83909 + y(-8.117954 + 1.8593222 y))))), \quad y = \sqrt{1 - \omega_0} \equiv \sqrt{b \ c},$$
(49)

An empirical representation of the Petzold's experimental angular scattering coefficient $\beta(\theta)$ and of the scattering phase function $p(\theta)$, where θ is the scattering angle in degrees, are represented by the following equations (Haltrin, 1997):

$$\beta(\theta) = \exp\left[q\left(1 + \sum_{n=1}^{5} (-1)^n k_n \theta^{\frac{n}{2}}\right)\right],\tag{50}$$

$$q = 2.598 + 17.748\sqrt{b} - 16.722b + 5.932b\sqrt{b} , \qquad (51)$$

$$k_{1} = 1.188 - 0.688 \omega_{0},$$

$$k_{2} = 0.1(3.07 - 1.90 \omega_{0}),$$

$$k_{3} = 0.01(4.58 - 3.02 \omega_{0}),$$

$$k_{4} = 0.001(3.24 - 2.25 \omega_{0}),$$

$$k_{5} = 0.0001(0.84 - 0.61 \omega_{0}),$$
(52)

$$p(\theta) = \frac{4\pi}{b} \exp\left[q\left(1 + \sum_{n=1}^{5} (-1)^n k_n \theta^{\frac{n}{2}}\right)\right]$$
 (53)

The strong regressions given by Eqs. (50)-(53) can be used as a basis for the empirical model of the phase functions with the coefficients dependent on the absorption and scattering coefficients. The single-scattering albedo used here varies from 0.09 to 0.96.

EXAMPLES OF COMPUTATIONS

To illustrate presented theory and algorithm, several examples of the sea shadowed by a long pier are presented here. All calculations are made with the FORTRAN program RADPIER listed in Appendix A. Below we present one line graph (Fig. 2) that illustrates the range of changes in the reflected radiance inflicted by the pier and six density graphs (Figs.3-8) that illustrate the two-dimensional pictures of reflected radiances.

The one-dimensional graph presented by Fig. 2. shows radiance distribution of reflected light over 0x axis in the direction perpendicular to the pier. The values of inherent optical properties are taken as follows; $c = 0.5 \, m^{-1}$, $b = 0.2 \, m^{-1}$; the solar zenith angle is taken to be 45°, and the solar azimuth angle is equal to 0° (the solar plane is orthogonal to the pier direction); the viewing zenith angle is set to 30°. Here we considered two viewing cases: (1) viewing azimuth angle equal to 15°, and (2) viewing azimuth angle is equal to 105°. In Fig. 2 we see a very sharp shadow with a peculiar spatial behavior that is a result of backscattering from the sea. Zero values of radiance in this picture are results of neglecting the part of illumination that is due to the diffuse light of the sky.

Next two figures (Fig. 3 and Fig. 4) show two-dimensional density plots of normalized sea radiance as a function of coordinate *x* and azimuth angle of viewing. Zenith angle of viewing is 30°, solar zenith angle is 45°. Fig. 3 displays the radiance with progressive gray-scale palette and Fig. 4 displays the same values as Fig. 3 only plotted with the banded gray-scale palette to show a fine structure of the light field.


The next two figures (Fig. 5 and Fig. 6) display two-dimensional density plot of normalized sea radiance as a function of coordinate x and azimuth angle of viewing. Zenith angle of viewing is 35° , solar zenith angle is 45° . Fig. 5 displays the radiance with progressive gray-scale palette and Fig. 6 displays the same values as Fig. 5 only plotted with the banded gray-scale palette to show the fine structure of the light field.

The last two figures (Fig. 7 and Fig. 8) are similar to the previous figures only for different angles: zenith angle of viewing is 75°, and solar zenith angle equal to 45°. The difference in figures 7 and 8 are also in palettes: Fig. 7 is drawn with progressive gray-scale palette, and Fig. 8 is drawn with the banded gray-scale palette to show a fine structure of the light field.

CONCLUSIONS

The equations (44)-(47) expresses the upwelling sea radiance distribution as a function of solar zenith angle, azimuth and zenith angles of viewing, the width of the pier and the inherent optical properties of the water (absorption and attenuation coefficients and the phase function of scattering). Results for the actual radiance distribution of light are obtained numerically using the *in situ* measured optical properties of seawater and are shown in Figs. 3 – 8. The scattering phase function used in these calculation is estimated through the regression equations (50)–(53) that are derived from an analysis of all fifteen Petzold scattering phase functions. The derived scattering phase function depends the scattering coefficient, the single-scattering albedo, and the scattering angle.

The main final result of this report consists of the Eqs. (33)-(35) for the upwelling sea angular radiance distribution as a function of: 1) the solar zenith angle, 2) azimuth and zenith angles of viewing, 3) the width of the opaque body, 3) absorption and attenuation coefficients, and 4) the phase function of scattering.






Fig. 3. Two-dimensional density plot of normalized sea radiance as a function of coordinate x and azimuth angle of viewing. Zenith angle of viewing is 30° , solar zenith angle is 45° .




Fig. 4. Two-dimensional density plot of normalized sea radiance as a function of coordinate *x* and azimuth angle of viewing. Zenith angle of viewing is 30°, solar zenith angle is 45°. The same as Fig. 3 only plotted with the banded gray-scale palette to show the fine structure of the light field.




Fig. 5. Two-dimensional density plot of normalized sea radiance as a function of coordinate x and the azimuth angle of viewing. Zenith angle of viewing is 35° , solar zenith angle is 45° .




Fig. 6. Two-dimensional density plot of normalized sea radiance as a function of coordinate *x* and azimuth angle of viewing. Zenith angle of viewing is 35°, solar zenith angle is 45°. The same as Fig. 5 only plotted with the banded gray-scale palette to show the fine structure of the light field.




Fig. 7. Two-dimensional density plot of normalized sea radiance as a function of coordinate x and azimuth angle of viewing. Zenith angle of viewing is 75°, solar zenith angle is 45°.




Fig. 8. Two-dimensional density plot of normalized sea radiance as a function of coordinate *x* and azimuth angle of viewing. Zenith angle of viewing is 75°, solar zenith angle is 45°. The same as Fig. 7 only plotted with the banded gray-scale palette to show the fine structure of the light field.

ACKNOWLEDGMENTS

The author is grateful to Dr. Peter M. Smith of NRL Code 7340 for bringing attention to this problem, and Drs. Albert W. Green and Alan D. Weidemann of NRL Code 7330 for useful help. This work was funded through the Littoral Optical Environment Program (LOE 6640-07 and LOE 6640-08) and the cooperative task No. 5972-A7 with the NRL Code 7240 (now 7340). This report represents NRL contribution PU/7331-98-0340.

This page intentionally left blank

REFERENCES

- S. Chandrasekhar, Radiative Transfer, Dover Publications, New York, pp. 393, 1960.
- V. V. Vladimirov, "Mathematical problems of the one-velocity theory of particle transport," in: *Proc. V. A. Steklov Inst. of Math.*, Vol. 61, p. 3-158, 1961, in Russian.
- H. G. Kaper, "Elementary solutions of the reduced three-dimensional transport equation," *J. Math. Phys.*, Vol. 10, No. 2, pp. 286-297, 1969.
- K. M. Case and R. D. Hazeltine, "Three-Dimensional Linear transport theory," *J. Math. Phys.*, Vol. 11, No. 4, pp. 1126-1135, 1970.
- V. M. Kochetkov, "Fundamental solution of an unsteady transport equation in solid geometry," *Appl. Math. Mech. (PMM)*, Vol. 32, No. 3, pp. 393-400, 1968.
- P. J. Flateau and G. L. Stephens, "On the Fundamental solution of the radiative transfer equation," *J. Geophys. Res.*, Vol. 93, No. D9, pp. 11037-11050, 1988.
- M. M. Williams, "Approximate solutions of the neutron transport equation in two and three dimensional systems," *Nucleonic*, Vol. 9, No. 7, pp. 305-312, 1967.
- N. G. Jerlov, Marine Optics, Elsevier Press, Amsterdam, pp. 231, 1986.
- C. D. Mobley, *Light and Water*, Academic Press, San Diego Toronto, 592 p.,1994.
- V. I. Haltrin (a.k.a. В. И. Халтурин), "Propagation of Light in a Sea Depth," Chapter 2 in: Remote Sensing of the Sea and the Influence of the Atmosphere, Moscow-Berlin-Sevastopol, Publ. by the German Democratic Republic Acad. of Sci. Inst. for Space Research, pp. 20-62, 1985, in Russian.
- V. I. Haltrin and G. W. Kattawar "Self-consistent solutions to the equation of transfer with elastic and inelastic scattering in oceanic optics: I. Model," *Appl. Optics*, Vol. 32, No. 27, pp. 5356-5367, 1993.
- V. I. Haltrin, "Self-consistent approach to the solution of the light transfer problem for irradiances in marine waters with arbitrary turbidity, depth and surface illumination: I. Case of absorption and elastic scattering," *Appl. Optics*, Vol. 37, pp. 3773-3784, 1998a.
- V. I. Haltrin, "Theoretical and empirical phase functions for Monte Carlo calculations of light scattering in seawater," in *Proceedings of the Fourth International Conference Remote Sensing for Marine and Coastal Environments: Technology and Applications*, Vol. I, Publ. by Envir. Res. Inst. of Michigan, Ann Arbor, MI, pp. 509-518, 1997.
- V. I. Haltrin, "Empirical relationship between aerosol scattering phase function and optical thickness of atmosphere above the ocean," in *SPIE*, Vol. 3433, paper No. 39, 1998b (to be published).
- V. I. Haltrin, "Algorithm for Atmospheric Correction of Airborne AVIRIS Ocean Images," in *SPIE*, Vol. 3433, paper No. 41, 1998c (to be published).

- P. M. Morse and H. Feshbach, *Methods of Theoretical Physics*, Part 1 and 2, McGraw Hill Book Co., New York-Toronto-London, pp. 1978, 1953.
- V. V. Vladimirov, Equations of Mathematical Physics, Dekker, New York, pp. 418, 1971.
- K. S. Shifrin, *Physical Optics of Ocean Water*, American Institute of Physics, New York, pp. 285, 1988.
- V. V. Shuleykin, *Physics of the Sea (Fizika Morya)*, Nauka, Moscow, pp. 1083, 1968, in Russian.
- V. A. Timofeyeva, "Optical characteristics of turbid media of the seawater type," *Izv., Atm. Ocean. Physics*, Vol. 7, No.12, pp.1326-1329 (English translation: pp. 863-865), 1971.

Appendix A

FORTRAN PROGRAM RADPIER

```
*******************
                         program RadPier
C
 ********************
         Created May 5, 1997. Last modification: May 9, 1997
            Written by Vladimir I. Haltrin, NRL Code 7331
С
        Ph.: (228) 688-4528, e-mail: <haltrin@nrlssc.navy.mil>
C
C
                   This code can be downloaded
C
     from an anonymous ftp site at <serv7330.nrlssc.navy.mil>
C
 ******************
С
     This program calculates an x-profile of radiance reflected by the
С
     homogeneous sea illuminated by the direct sun light and shadowed
C
     by the pier of 2w width (in 0x direction) and infinitely stretched
С
     in the Oy direction. The depth is measured along the Oz axis.
C
     The radiance L is a function of two viewing angles that and fi,
C
     and horizontal coordinate x (which is perpendicular to the pier).
С
С
     L is also parametrically dependent on the inherent optical
C
     parameters of water (b, c), the sun zenith angle zang, and the
     halfwidth of the pier w.
С
C
       This program is based on the two-dimentional in space (x, z)
C
     theoretical algorithm derived in April 1997 by V.I. Haltrin
С
     specifically for the cooperative task No. 5979-A7 with the NRL
С
     Code 7240 (now Code 7340) (P.I. Dr. Peter M. Smith).
С
       The algorithm is based on the scalar radiative transfer theory.
С
     It uses Green's function approach to solve the problem in
C
     z-direction and the Fourier-transform approach to solve the
С
     problem in x-direction. In some integrals in order to receive
С
     simple analytical solutions the phase-functions are approximated
С
     by the transport phase function.
С
       It is possible to rewrite this code without such approximation;
С
     but in that case it will incorporate two-dimentional numerical
C
     integrations.
     implicit none
               i,j,M,N,Nx,Na
     integer
     parameter (N=501,M=181)
     real*8 x(N),L(N,M),fia(M)
     real*8
              a,c,w,zang,thet,fi,x0,dx
     real*8
              b, omeg, bp, bb, mua, mus, Rb, R0, theta, mu, xx
              psip, fsip, angd, fourpi, phf, mux, fis
     real*8
     real*8
              F1,F2,tau,op,L0,rL11,xi,dfi
            muav, muwa, xgor, fpsip, cosgam, pphf, ftau! functions
     real*8
     character tb
               fourpi /12.5663706/
     data
```

```
open(11, file='radpier.in', status='old')
   read(11,*) b     ! water scattering coefficient in 1/m
   read(11,*) c    ! water attenuation co
read(11,*) w    ! pier halfwidth in m
                     ! water attenuation coefficient in 1/m
   read(11,*) zang ! sun zenith angle in degrees
   read(11,*) thet ! viewing zenith angle in degrees
   read(11,*) Na ! Number of viewing azimuth angles in degrees read(11,*) Nx ! size of x-array (Nx \leq N) read(11,*) x0 ! initial x coordinate in m read(11,*) dx ! spatial step in positive 0x direction in m
close(11)
a = c-b
                      ! absorption coefficient
omeg = b/c
                      ! single-scattering albedo
mua = muav(omeg)
                     ! average cosine
                     ! cosine of the sun zenith angle in water
mus = muwa(zang)
mu = muwa(thet) ! cosine of zenith viewing angle in water
xx = xgor(mua) ! Gordon's coefficient x=bb/(a+bb)
bb = a*xx/(1.-xx) ! backscattering coefficient
bp = bb/b ! backscattering probability
bp = bb/b
                           ! backscattering probability
psip = fpsip(bp,mu,mus) ! auxiliary function of the theory
L0 = mus*omeg/fourpi
L0 = L0/(mus+mu)
op = 0.5*omeg*psip
fis = 0.
                       ! the sun plane is orthogonal to the pier
dfi = 180./(Na-1) ! increment in viewing azimuth angle
do j = 1, Na
   fi = dfi*(j-1)
   fia(i) = fi
                                    ! viewing azimuth angle
   mux = cosgam(mu, mus, fi, fis) ! cos of direct sun rays in water
   angd = ACOSD(mux)
   phf = pphf(b,omeg,angd) ! scattering phase function
   tau = ftau(c,mu,fi)
   do i = 1, Nx
      xi = x0+dx*(i-1)
      x(i) = xi
      call ff12(w,xi,tau, F1,F2)
      L(i,j) = L0*((1.-F1)*phf+(1.-F2)*op)! Total radiance
   end do
end do
rL11 = 1./L(1,1)
tb = CHAR(9)
open(21, file='radpier.inf', status='new')
   write(21,55) 'Distribution of radiance reflected by the sea'
   write(21,55) 'shadowed by long pier oriented along axis 0y.'
   write(21,55) '-----'
   write(21,66) 'water scattering coefficient in 1/m, b=',b
   write(21,66) 'water attenuation coefficient in 1/m, c=',c
   write(21,66) 'pier halfwidth in meters,
                                                              w=', w
   write(21,66) 'sun zenith angle in degrees,
                                                          zang=',zang
   write(21,66) 'viewing zenith angle in degrees, thet=',thet
                                   phf = pphf(b,omeg,angd) = ',phf
   write(21,66) 'Function
   write(21,66) 'Function
                                          op = 0.5*omeg*psip = ',op
   write(21,66) 'initial x coordinate in meters, x0=',x0
   write(21,66) 'spatial step in positive 0x direction, dx=',dx
close(21)
```

```
open(22, file='radpier.out', status='new',recl=1800)
      write(22,73) Nx,tb,Na
      write(22,73) 0.,tb,0.
      write(22,75) x(1),(tb, x(i), i=2,Nx)
      write(22,77) fia(1), (tb, fia(j), j=2, Na)
      do i = 1, Nx
        write(22,88) rL11*L(i,1),(tb,rL11*L(i,j),j=2,Na)
      end do
    close(22)
  55 format(a45)
  66 format(a42,f8.3)
  73 format(i4,a1,i4)
  75 format(f8.2,200(a1,f8.2))
  77 format(f8.2,180(a1,f8.2))
  88 format(f8.6,180(a1,f8.6))
    end
real*8 function muav(omega)
This regression is derived by the author (V.I.Haltrin) from the
С
   experimental results by V. A. Timofeyeva, "Optical characteristics
С
   of turbid media of the seawater type, "Izv., Atm. Ocean. Physics,
   Vol.7, No.12, pp.1326-1329 (English translation: pp.863-865), 1971.
implicit
           none
    real*8
            omega, y
    y = SQRT(1.-omega)
    muav = y*(2.6178398+y*(-4.6024180+y*(9.0040600+
         y*(-14.59994+y*(14.83909+y*(-8.117954+1.8593222*y))))))
    return
    end
real*8 function muwa(thet)
Here a text-book formula is coded.
none
    implicit
    real*8
           thet,z,nw
           nw / 1.34 /
    data
    z = SIND(thet)/nw
    muwa = SQRT(1.-z*z)
    return
    end
```

```
real*8 function xgor(mua)
 ********************
    Formula used here is taken from the book chapter by V.I. Haltrin:
C
    "Propagation of Light in a Sea Depth," Ch. 2 in: Remote Sensing of
С
    the Sea and the Influence of the Atmosphere (in Russian), Moscow-
C
    Berlin-Sevastopol, Publ. by the GDR Academy of Sciences Institute
С
    for Space Research, pp. 20-62, 1985.
implicit
            none
    real*8
           mua,mu2,d
    mu2 = mua*mua
    d = 1.-mu2
    d = d*d
    d = d/(1.+mu2*(4.-mu2))
    xqor = d
    return
    end
real*8 function fpsip(bp,mu,mus)
 *******************
    Formula used here is derived by V.I. Haltrin specifically for the
C
    task No. 5979-A7 for the NRL code 7240 (P.I. Peter Smith)
implicit
            none
    real*8
            bp,mu,mus,lm,fp,ps,dm
    lm = LOG(1.+ 1./mu)
    lm = mu*lm
    fp = 1.-bp
    dm = fp-bp
    ps = bp*fp*lm
    ps = ps+(bp*fp*mus+dm*dm*mu)/(mus+mu)
    fpsip = 4.*ps
    return
    end
 ********************
    real*8 function pphf(b,omeg,angd)
 ******************
С
      The equations for this function are taken from: V. I. Haltrin,
C
    "Theoretical and empirical phase-functions for Monte-Carlo calcu-
C
    lations of light scattering in sea water." in: Proceedings of the
С
    Fourth International Conference Remote Sensing for Marine and
С
    Coastal Environments: Technology and Applications," Vol. I,
C
    Publication by Environmental Research Institute of Michigan,
С
    Ann Arbor, Michigan, USA, pp. 509-518, 1997.
C
      These equations approximate all 15 Petzold phase functions
C
    and relate them with the scattering coefficient b and single-
С
    scattering albedo omeg; angd is the scattering angle in degrees.
С
```

```
implicit none
    real*8 b,omeg,angd,sa,sb,q
    real*8
          k1,k2,k3,k4,k5,sm,bt,pi
          pi /3.14159265/
    sa = SQRT(angd)
    sb = SQRT(b)
    q = 2.598 + sb*(17.748 + sb*(-16.722 + 5.932*sb))
    k1 = 1.188 - 0.688 * omeg
    k2 = 0.1*(3.07-1.9*omeg)
   k3 = 0.01*(4.58-3.02*omeg)
   k4 = 0.001*(3.24-2.25*omeg)
   k5 = 0.0001*(0.84-0.61*omeg)
    sm = sa*(-k1+sa*(k2+sa*(-k3+sa*(k4-k5*sa))))
   bt = EXP(q*(1.+sm))
   pphf = (4.*pi*bt)/b
   return
    end
real*8 function cosgam(mu, mup, fi, fip)
Here a text-book formula is coded.
implicit none
    real*8
          mu, mup, fi, fip, cg
    cg = COSD(fi-fip)
    cg = cg*SQRT(1.-mup*mup)
    cg = cg*SQRT(1.-mu*mu)
    cg = cg + mu * mup
    cosgam = cg
    return
    end
real*8 function ftau(c,mu,fi)
 **********************
      This is auxilary function for the subroutine ff12
implicit none
    real*8
          c,mu,fi,tau
    tau = SQRT(1.-mu*mu)
    tau = tau*COSD(fi)
    tau = tau/c
    if (tau.eq.0) tau = 1.e-99
    ftau = tau
    return
    end
```

```
subroutine ff12(w,x,tau, f1,f2)
Formulae used here is derived by V.I. Haltrin specifically for
    the task No. 5979-A7 for the NRL Code 7240 (P.I. Pit Smith)
implicit none
   real*8
          w,x,tau,f1,f2, w1,w2,s1,s2,st,e1,e2,a1,a2,at,d1,d2
   w1 = w-x
   w2 = w+x
    s1 = 1.
   if (w1.le.0.) s1 = -1.
   s2 = 1.
   if (w2.le.0.) s2 = -1.
   st = 1.
   if (tau.le.0.) st = -1.
   e1 = ABS(w1/tau)
   e1 = EXP(-e1)
   e2 = ABS(w2/tau)
    e2 = EXP(-e2)
   d1 = s1*(1.-e1)+s2*(1-e2)+st*(e2-e1)
   f1 = 0.5*d1
   a1 = ABS(w1)
   a2 = ABS(w2)
   d1 = (a1*e1-a2*e2)/tau
   at = ABS(tau)
   d2 = (w1*e1+w2*e2)/at
   f2 = f1 + 0.5*(d1-d2)
   return
    end
```

A sample of an input file 'radpier.in' (starts just after the dotted line):

```
0.2 b = water scattering coefficient in 1/m
0.5 c = water attenuation coefficient in 1/m
1.5 w = pier halfwidth in meters
45. zang = sun zenith angle in degrees
25. thet = viewing zenith angle in degrees
181 Na = number of viewing azimuth angles in degrees (O-\pi)
```

201 Nx = size of x-array (Nx \leq N) -10. x0 = initial x coordinate in meters 0.1 dx = spatial step in the positive 0x direction in m

Distribution List:

Maury Oceanographic Library Naval Oceanographic Office 1002 Balch Blvd. Stennis Space Center MS 39522-5001

Naval Research Lab. Library Washington DC 20375

Library of Congress Washington, DC

TAMU Library Texas A&M University College Station, Texas 77843

Dr. Peter M. Smith NRL Code 7442 Stennis Space Center, MS 39529 ext 5267 smith@nrlssc.navy.mil

Dr. Vladimir I. Haltrin NRL Code 7331 Stennis Space Center, MS 39529 ext 4528 haltrin@nrlssc.navy.mil

Dr. Todd Holland NRL Code 7442 Stennis Space Center, MS 39529 ext 4866

Dr. Alan D. Weidemann NRL Code 7331 Stennis Space Center, MS 39529 ext 5253 alanw@nrlssc.navy.mil 25 copies

Dr. Steven G. Ackleson ONR 3220P, Office of Naval Research 800 N. Quincy Street Arlington, VA, 22043 ACKLESS@ONR.NAVY.MIL

Dr. Curtiss O. Davis Code 7212 Naval Research Laboratory 4555 Overlook Ave. S. W. Washington DC 20375 davis@rsd.nrl.navy.mil

Dr. Howard, R. Gordon University of Miami, Department of Physics, P. O. Box 248046 Coral Gables, FL 33124-8046 gordon@phyvax.ir.miami.edu

Dr. Norman J. McCormick University of Washington Department of Mechanical Engineering Box 352600 Seattle, WA 98195-2600 mccor@u.washington.edu

Prof. Kusiel S. Shifrin College of Ocean and Atmospheric Sciences Oregon State University Oceanology Administration Bld 104 Corvallis, Oregon 97331-5503 shifrink@ucs.orst.edu

Dr. Kenneth Voss University of Miami, Department of Physics, P. O. Box 248046 Coral Gables, Florida 33124 kvoss@umiami.miami.edu

Dr. Robert H. Stavn Department of Biology University of North Carolina / Greensboro 1000 Spring Garden Street, Greensboro, NC 27412 stavnrh@iris.uncg.edu Dr. Curtis D. Mobley Sequoia Scientific, Inc. 9725 SE 36th Street, Ste 308 Mercer Island, WA 98040 mobley@sequoiasci.com

Dr. George W. Kattawar Department of Physics Texas A&M University College Station, Texas 77843-4242 kattawar@tamu.edu

Dr. Robert A. Maffione HobiLabs 55 Penny Lane, Suite 104 Watsonville, CA 95076 maffione@hobilabs.com

Dr. Robert Holman College of Oceanography Oregon State University Ocean Admin Building 102 Corvalis, OR 97331-5503 holman@oce.orst.edu 541-737-2914

Dr. David Lyzenga University of Michigan Dept. Naval Arcitecture and Engineering Ann Arbor, MI 48109-2145 lyzenga@umich.edu

Dr. Charles R. Bostater Oceanography, Ocean Engineering and Environmental Science Florida Institute of Technology 150 West University Boulevard Melbourne, Florida 32901-6988